/*

Stephen Morgan	
Proposal title: A Question-Wording Experiment on Support for Free Expression

HYPOTHESES

Stated-Hyp1: The primary hypothesis is that respondents will extend fewer opportunities
for free expression to the “Islamic cleric” vs. Muslim clergyman

	Test-Hyp1: Support for free speech will be lower when term Islamic cleric is used vs. Muslim clergyman.

Stated-Hyp2: The term “religious leader” should yield an intermediate result.

	Test-Hyp2: Support for free speech will be lower when term Islamic religious leader is used vs. Muslim clergyman.

	Test-Hyp3: Support for free speech will be higher when term Islamic religious leader is used vs. Islamic cleric.
	
********************************************************************************
NOTES

- there are 3 outcome variables with similar response categories. We are 
combining these into one summated index of support for free speech.

*/

clear all
use "MorganS30.dta", clear

********************************************************************************

* CONSTRUCT INDICATORS OF EXPERIMENTAL MANIPULATIONS

* term used for Islamic cleric
	tab P_CONDITION
	gen treat= P_CONDITION
	lab def term 1 "Muslim clergyman" 2 "Islamic cleric" 3 "Islamic religious leader"
	lab val treat term
	tab treat, mis
	
* CONSTRUCT OUTCOME MEASURES

* allow speech in community?
	gen allow_speech=.
	replace allow_speech=1 if Q1==1| Q3A==1| Q7==1
	replace allow_speech=0 if Q1==2| Q3A==2| Q7==2
	tab allow_speech
	
* allow teaching in college
	gen allow_teach=.
	replace allow_teach=1 if Q2==1|Q5==1|Q8==1
	replace allow_teach=0 if Q2==2|Q5==2|Q8==2
	tab allow_teach
	
* allow book in library (if R favors removing book, then allow book=0)
	gen allow_book=.
	replace allow_book=1 if Q3==2 |Q6==2| Q9==2
	replace allow_book=0 if Q3==1 |Q6==1| Q9==1
	tab allow_book
	
* support for free speech scale
	gen support_freespeech = allow_speech + allow_teach + allow_book
	tab support_freespeech

********************************************************************************

* TEST HYPOTHESIS 
	
*Test-Hyp1: Support for free speech will be lower when term Islamic cleric is used vs. Muslim clergyman.
	reg support_freespeech i.treat if treat!=3
	// do not reject. 0.04
	tess 2.treat -, init(MorganS30)
	
*Test-Hyp2: Support for free speech will be lower when term Islamic religious leader is used vs. Muslim clergyman.
	reg support_freespeech i.treat if treat!=2
	// reject. 0.344
	tess 3.treat -
	
*Test-Hyp3: Support for free speech will be higher when term Islamic religious leader is used vs. Islamic cleric.
	reg support_freespeech i.treat if treat!=1
	// reject. 0.277
	tess 3.treat +
	
